

Following is a response from DHS Senior Research Associate, Shireen Assaf:

Thank you for your question. The Stata code below will provide you with the correct estimates of vaccination coverage shown on page 140 for either source with the correct denominator of 1590 children. Note that for DPT there were the standard DHS variables of h3, h5, and h8 but also country specific variables of sdh1, sdh2, sdh3.

**** Stata code for vaccination coverage in Uganda 2006 survey ****

```
gen age = v008 - b3

gen dpt=0
replace dpt=1 if h3>0 & h3<8 & h5>0 & h5<8 & h7>0 & h7<8
replace dpt=1 if sdh1>0 & sdh1<8 & sdh2>0 & sdh2<8 & sdh3>0 & sdh3<8
replace dpt=. if age<12 | age>23 | b5==0

gen measles=(h9==1|h9==2|h9==3)
replace measles = . if age<12 | age>23 | b5==0

gen polio=0
replace polio=1 if h4>0 & h4<8 & h6>0 & h6<8 & h8>0 & h8<8
replace polio = . if age<12 | age>23 | b5==0

gen bcg=0
replace bcg=1 if h2>0 & h2<8
replace bcg = . if age<12 | age>23 | b5==0

gen vaccine=0
replace vaccine=1 if polio==1 & dpt==1 & measles==1 & bcg==1
replace vaccine = . if age<12 | age>23 | b5==0

gen wt= v005/1000000

*
tab bcg [iw=wt]
*matches 90.5% for BCG in report
tab dpt [iw=wt]
*matches 63.9% for all three doses of DPT in report
tab measles [iw=wt]
*matches 68.1% for measles in report
tab polio [iw=wt]
```

*matches 59.3% for all three doses of polio in report
tab vaccine [iw=wt]
*matches 46.2% for all basic vaccinations in report
*/
